
INTRODUCCIÓN AL MÓDULO PROBACKTEST

ProBacktest es la herramienta de la Plataforma Trader Pro 8.02 para crear sistemas de inversión personaliza-
dos y probar su e�cacia a lo largo de cualquier período del histórico disponible para un valor dado.

ProBacktest integra el lenguaje de programación ProBuilder (le aconsejamos la lectura previa del manual Pro-
Builder) con extensiones aplicables exclusivamente en el desarrollo de sistemas.

En este módulo puede simular aperturas de posiciones en función de condiciones personalizadas queintegran:

�� Posicionamiento en el mercado (orden pendiente o no,…).
�� Fechas de posicionamiento en el mercado (abrir la posición en la vela del día siguiente,…).
�� Modalidades de posicionamiento (apertura de posición a precio de mercado, a precio límite X,…).
�� Posicionamiento de stops.
�� Resultados.
�� Precio de ejecución de la última orden.

Los resultados de una simulación ProBacktest se presentan a través de los siguientes elementos:

�� Curva de ganancias y pérdidas (o ‘Equity Curve’), que indica el estado de su cartera virtual a medida que 
se aplica el sistema.
�� Histograma de posiciones (verde en compras, rojo en ventas a descubierto) de liquidación y de zonas sin 
transacciones en curso (sin velas).
�� Informe detallado, que ofrece estadísticas generales de su sistema para el valor, período e históricos elegidos.

Este documento es una extensión del manual ProBuilder pero también puede leerse de manera aislada.

Los usuarios más familiarizados con la programación pueden pasar directamente a la lectura del capítulo II o 
consultar el glosario para localizar rápidamente la explicación detallada de la función que busquen.

El contenido de todo el manual está dirigido a enseñarle a desarrollar y a probar sus propias ideas. En ningún 
caso se busca proporcionarle consejos de inversión.

Le deseamos una agradable lectura y los mejores resultados en sus inversiones.

CAPÍTULO I: INTRODUCCIÓN

EL ACCESO AL MÓDULO PROBACKTEST

El espacio de programación ProBacktest es accesible a partir del botón ‘Indicador/Backtest’ situado en la zona 
superior derecha de cualquier ventana de grá�co de su plataforma Trader Pro8.02.

Tras pulsar ese botón, se abrirá por defecto la ventana de gestión en la pestaña ‘Indicadores’. Active la pesta-
ña ‘ProBacktest’ pulsando sobre la misma con el ratón. Con ello accederá a:

�� La lista de los ProBacktests ya existentes (predefinidos o personalizados).
�� Los controles de creación de un nuevo ProBacktest.
�� Modificar o suprimir un ProBacktest existente.

Para crear un nuevo ProBacktest, pulse ‘Nuevo ProBacktest’ y elija entre la posibilidad de crearlo mediante 
programación (‘Creación programando’) o con el asistente a la programación (‘Creación asistida’).

ZONAS DE CONFIGURACIÓN DEPROBACKTEST

Comenzaremos por examinar la creación de un backtest mediante programación. La ventana  se compone de 
5 zonas con�gurables:

�� Zona de programación.
�� Gestión de capital (money management).

�� Optimización de variables.
�� Fecha de inicio y de final.
�� Visualización de resultados en gráfico y gestión de la ventana.

La primera sección le permitirá:

�� Programar directamente un sistema en la zona de texto.
�� Utilizar la función de ayuda ‘Insertar función’ y acceder a la biblioteca de las funciones disponibles.

La biblioteca de funciones las agrupa en 9 categorías distintas según el contexto en el que se aplican en la 
programación. Cada una de estas funciones se acompaña igualmente de un texto de ayuda.

En nuestro caso utilizaremos la biblioteca pulsando ‘Insertar función’.

Diríjase a la categoría ‘Comandos ProBacktest’ y pulse en ‘BUY’. A continuación, pulse el botón ‘Añadir’.

El comando se añadirá en la zona de programación.

Intentemos crear un backtest. Supongamos por ejemplo que deseamos comprar 10 valores a precio de mercado.

Para ello, realizaremos la operación descrita anteriormente para recuperar sucesivamente las instrucciones 
‘SHARES’, ‘AT’ y ‘MARKET’ (recuerde separar cada palabra de un espacio). La cantidad de valores a adquirir 
deberá insertarse entre los comandos ‘BUY’ y ‘SHARES’. Por último, deberá atribuir un nombre a su sistema 
(en este ejemplo, se le denomina ‘Mi Estrategia’.

La segunda sección (Money management) permite de�nir la gestión del capital: coste de cada transacción,  
capital a invertir y stops de protección asociados.

�� En ‘Gestión capital’ definiremos el capital inicial: la suma (ficticia) que deseamos invertir según el sistema 
que estamos creando, las comisiones de operativa, la gestión de riesgos y de posiciones.

�� En ‘Stops’ podemos seleccionar cualquiera de estos 4 tipos de stop: de protección (stop loss), de beneficio 
(stop profit, take profit o target), dinámico (trailing stop) y de inactividad.

Para más información acerca del ajuste de la gestión de capital, le sugerimos dirigirse al anexo disponible al 
�nal de este manual.

La tercera sección le permite de�nir la optimización de variables. A través de esta función, podrá probar múl-
tiples combinaciones de variables para encontrar aquellas que ofrezcan los mejores resultados.

El resultado de la optimización se presenta en un ‘Informe de optimización’. En el se indican las estadísticas 
de cada valor para determinar así la combinación de variables a utilizar en la optimización de su sistema.

He aquí un ejemplo de sistema con optimización de los períodos de medias móviles n y m:

Indicator 1 = Momentum [n] (Close) 
Indicator 2 = Average [m] (Indicator1)
IF Indicator1 CROSSES OVER Indicator 2 THEN
 BUY 1 SHARES AT MARKET THISBARONCLOSE
END IF
IF Indicator 1 CROSSES UNDER Indicator 2 THEN
 SELLSHORT 1 SHARES AT MARKET THISBARONCLOSE
END IF
 
De�nimos las variables optimizadas n y m de este modo:

�� ‘Nombre usado en programa’ representa el nombre que toma la variable en nuestro código (n y m en este 
caso). Es importante distinguir entre mayúsculas y minúsculas.
�� ‘Etiqueta en ventana propiedades’ representa el nombre que se le atribuye a la variable para que ésta sea 
reconocible con mayor facilidad (por ejemplo ‘Número de Períodos’ para n).
�� ‘Valor mínimo y ‘Valor máximo’ representan los extremos entre los que la variable puede oscilar durante 
las pruebas de optimización.
�� ‘Paso’ representa el intervalo de valores que la optimización respetará durante el análisis de los resultados.

He aquí un ejemplo de informe de optimización:

El informe del ejemplo ofrece 6 estadísticas para cada una de las combinaciones de variables estudiadas (en 
este caso: n y m). Veamos las estadísticas con más detalle:

�� Beneficios netos, designa la plusvalía obtenida con la operativa realizada. Matemáticamente, se traduce 
en esta fórmula:

Bene�cios netos = Valor del capital tras operativa realizada – Valor inicial del capital

Esta estadística permite evaluar de manera absoluta el potencial de ganancias con el sistema de�nido para 
cada valor de variables correspondientes.

Nota: las comisiones de operativa de�nidas en la sección ‘Money management’ se toman en cuenta en este 
cálculo.

�� Retorno de capital es el beneficio expresado en %. Matemáticamente, se traduce por la fórmula:

Retorno de capital = (100 x Bene�cios netos) / Valor inicial del capital

Indica el resultado relativo de este sistema, con�gurado con los valores respectivos de  las variables.

�� Max drawdown, indica la pérdida máxima incurrida al aplicar el sistema. Dicho de otro modo, es la dife-
rencia entre el punto máximo y el mínimo en la curva de pérdidas y ganancias.

Veamos un ejemplo de esta estadística:

El max drawdown es determinante dentro de su per�l de riesgo: si no desea exponerse a pérdidas importantes 
en una misma posición (incluso si el sistema en su conjunto pueda dar resultados muy positivos), deberá optar 
por una estrategia menos arriesgada.

�� Órdenes indica la cantidad de órdenes enviadas desde el lanzamiento de este sistema.

�� % Operaciones ganadoras indica el porcentaje de operaciones ganadoras dentrodel sistema, y constituye 
una información complementaria a Max drawdown. Matemáticamente se define así:

% Operaciones ganadoras = (100 x Núm.de operac. ganadoras) / Núm.total de operaciones

�� Esperanza es la media de ganancias por transacción y puede ayudar a determinar la eficacia media de las 
órdenes enviadas. La esperanza es particularmente importante cuando no deseamos enviar un gran número 
de órdenes; en tal caso, la estadística mencionada se convierte en un criterio determinante para decidir si 
el sistema se aplica o no.

Matemáticamente, se de�ne así:

Esperanza = Total neto / Número de transacciones

Nota: los valores óptimos de las variables de un sistema pueden cambiar para un mismo valor en función de 
las unidades de tiempo o el período histórico utilizados.

La cuarta sección sirve para de�nir el intervalo de tiempo sobre el que desee aplicar el sistema. Preste atención 
a la fecha �nal: en ella se cierran todas las posiciones que estén abiertas.

Esta función de ProBacktest se con�gura por defecto para probar el sistema a lo largo de la totalidad del his-
tórico visible (las posiciones abiertas sólo se cierran cuando se da la condición de salida).

RESULTADOS DE PROBACKTEST

Además del informe de optimización ya presentado, los resultados de un ProBacktest se ofrecen bajo 3 as-
pectos complementarios:

1. Grá�co liquidez

También llamada ‘Equity Curve’, representa la evolución del capital invertido (cuyo montante inicial se de�ne 
en la gestión de capital o money management) bajo el supuesto de que el sistema se haya aplicado desde el 
inicio del histórico visible en el valor del grá�co.

El grá�co de liquidez se muestra de color verde para indicar una variación positiva y rojo cuando la variación 
es a la baja.

2. Posiciones

Esta información muestra un histograma con la evolución de las posiciones abiertas a raíz del sistema.

�� Una vela verde indica la apertura de una posición a largo (compra).
�� Una vela roja indica la apertura de una posición a corto (venta a descubierto o short selling).
�� Si no hay ninguna vela visible, no hay ninguna posición abierta en el mercado.

La aparición de varias velas sucesivas de un mismo color indica que la(s) posición(es) son conservadas.

A lo largo del eje vertical visible en la zona derecha del grá�co se indica la cantidad de posiciones abiertas 
y acumuladas. En la ilustración siguiente constataremos que nos encontramos en una posición de venta a 
descubierto en 4 períodos.

3. Informe detallado

El informe detallado le ofrece una tabla con las principales estadísticas de su sistema: resultados, duración y listado 
de las posiciones abiertas. El informe detallado se presenta en una ventana independiente de tres pestañas:

�� En la pestaña Estadísticas hallará un presentación exhaustiva de los resultados del sistema (pérdidas y ganan-
cias, número de operaciones ganadoras,…). Además de las informaciones más conocidas, señalaremos la 
relevancia de la ‘Mayor ganancia’ y ‘Mayor pérdida’, así como el DrawDown (máxima pérdida sucesiva). El 
análisis de estos dos valores le permitirá evaluar si la estrategia analizada es compatible con su propio perfil 
de riesgo.

�� La pestaña ‘Lista de órdenes’ le indica la hora, el sentido, la cantidad y el precio de las órdenes enviadas. 
Estas órdenes aparecen en el huso horario del mercado (es decir, expresadas en hora local).

�� Por último, la pestaña 'Seguimiento posiciones’ le da información sobre las posiciones (a largo o a corto, 
duración indicada en cantidad de velas, resultados absolutos y relativos de cada posición, fecha de entrada 
y de salida…).

CAPÍTULO II: LA PROGRAMACIÓN DE PROBACKTEST

ENTRADA Y SALIDA DEL MERCADO

Vamos a ilustrar dos categorías de instrucciones que permiten entrar y salir del mercado: las instrucciones 
de apertura de posición y de cierre de posición(stop).

1. Instrucciones de apertura deposición

Las instrucciones se diferencian en función del sentido de la posición:

�� Posiciones largas

�� BUY instrucción de compra de valores (entrada a largo).
�� SELL instrucción de venta de valores adquiridos (salida de largo).

�� Posiciones cortas

�� SELLSHORT instrucción de ventea a descubierto de valores (entrada a corto).
�� EXITSHORT instrucción de compra de valores vendidos a descubierto (salida de corto).

ProBacktest no permite actualmente la simulación de ‘hedging’ (apertura simultánea de una posición y de otra 
en sentido inverso sobre un mismo valor). En la práctica, resulta posible cerrar una posición BUY con una 
SELLSHORT. No obstante, se aconseja respetar los binomios descritos para cerrar una posición, utilizando la 
correspondiente a cada apertura.

Todos estos comandos pueden acompañarse de los elementos que describimos a continuación:

SELLSHORT ‘Cantidad’ ‘Modo’ AT ‘Tipo’ ‘Fecha/hora de ejecución’

a. Cantidad
Consiste en el número de valores (volumen) que desea comprar o vender.

Atención: es posible no indicar ningún número. En tal caso, el programa considera por defecto la operativa 
con una acción o un lote en el caso de futuros).

b. Modo
Es posible de�nir el modo de inversión en términos absolutos o relativos.

�� SHARES acciones, futuros, warrants y Forex
�� CASH transacción en unidades monetarias (€ o $)
�� % CAPITAL transacción en porcentaje de capital (ex : 10% del capital)
�� % LIQUIDITY transacción en porcentaje de liquidez restante.

Ejemplo:

Cuando el RSI se halla en zona de sobreventa (RSI < 30) y el precio está situado por debajo  de la banda de 
Bollinger inferior, se genera una orden de compra a precio de mercado por un 10% del capital.

Cuando el RSI se halla en zona de sobrecompra (RSI > 70) y el precio está situado por debajo de la banda 
de Bollinger inferior, se genera una orden de venta en idénticas condiciones.

IF RSI [14] (Close) < 30 AND Close < BollingerDown [25] (Close) THEN
 BUY 10 %CAPITAL AT MARKET
END IF

IF RSI [14] (Close) > 70 AND Close > BollingerUp [25] (Close) THEN 
 SELL 10 %CAPITAL AT MARKET
END IF

c. Tipo

Dispone de tres tipos de ejecución de órdenes:

�� AT MARKET: La orden se envía a precio de mercado.
�� AT (price) LIMIT: Se ejecutará tras tocar el límite indicado y a precio que no lo sobrepase.
�� AT (price) STOP: La orden se ejecutará a precio de mercado tras tocar el stop indicado.

Ejemplo: Volatility Breakout

Este sistema se basa en la volatilidad. Se posiciona una orden de compra y una de venta a descubierto en 
cada vela, ambas a precio ’limite’.

La orden de compra se sitúa en el precio de cierre de la vela previa más un 50% del rango de la vela previa.

La orden de venta a descubierto se sitúa en el precio de cierre de la vela previa menos un 50% del rango de 
la vela previa.

REM Volatility Breakout
BuyLimit = Close [1] + (Range [1] * 50 /100) 
SellLimit = Close [1] - (Range [1] * 50 /100)
BUY 1 SHARES AT BuyLimit Stop
SELLSHORT 1 SHARES AT SellLimit Stop

d. Fecha/hora de ejecución

Por  defecto, cada orden se ejecuta al cierre de la vela actual. Por lo tanto, en la vela siguiente.

Sin embargo, es posible modi�car la fecha de ejecución en el caso de órdenes enviadas a precio de mercado 
a través de los comandos descritos a continuación (atención a su uso sin paréntesis ni corchetes):

�� NextBarOpen: crea la orden en la apertura de la vela siguiente (configuración por defecto).
�� NextBarClose: crea la orden en el cierre de la vela siguiente.
�� ThisBarOnClose: crea la orden en el cierre de la vela actual.
�� TodayOnClose: crea la orden en el cierre del día actual (relevante sólo si se emplea en intradía).
�� TomorrowOpen: crea la orden en la apertura del día siguiente (relevante sólo si se emplea en intradía).
�� TomorrowClose: crea la orden en el cierre del día siguiente (relevante sólo si se emplea en intradía).
�� RealTime: crea la orden en tiempo real, en el tick actual

Las instrucciones de tipo ‘Fecha/hora de ejecución’ sólo pueden emplearse cuando son precedidas de la 
instrucción ‘ATMARKET’.

Ejemplo: Rupturas de canal

El soporte y resistencia de un canal se re�eren a los valores mínimo y máximo de las dos últimas velas. Si antes 
de las 16h el precio rompe al alza la resistencia del canal, se genera  una orden de compra por un 70% del 
capital. La posición se cierra cuando el precio rompe a la baja el soporte del canal, siempre y cuando ello 
ocurra antes de las 16h.

REM Cierre de la segunda vela (índice 1) 
IF IntradayBarIndex = 1 THEN
 Resistencia = Highest [2] (High) 
 Soporte = Lowest [2] (Low)
END IF

REM Compra/Venta tras ruptura, si tiene lugar antes de las 16:00:00 (hora local) 
IF IntradayBarIndex > 1 AND Time < 160000 THEN

REM Ruptura de resistencia 
 IF Close > Resistencia THEN
  BUY 70 %CAPITAL AT MARKET THISBARONCLOSE
 END IF

REM Ruptura de soporte 
 IF Close < Soporte THEN
  SELLSHORT 70 %CAPITAL AT MARKET THIS BARONCLOSE 
 END IF
END IF

2. Instrucciones de STOP

Los Stops de un sistema pueden con�gurarse manualmente.

Además de los 4 tipos de stops disponibles en la sección 2 de la ventana de programación (ver sección ‘Mo-
ney Management’), también puede utilizar stops programados por usted mismo mediante el comando:

SET STOP (precio)

en el que la constante ‘precio’ indica el nivel en el que deseamos cerrar nuestra posición.

3. Estrategia de cruces precio/SAR

El siguiente sistema lanza una orden de compra/venta a descubierto en cuanto el precio cruza al alza/a la 
baja el SAR.

Además, se posiciona un stop dinámico de protección para cerrar la posición en caso de que el precio se 
aproxime a un nivel demasiado bajo (en el código, dicho stop se llama ’cut’).

Indicator 1 = Close
Indicator 2 = SAR [0.02,0.02,0.2]
 
REM Compra
c1 = (Indicator 1 Crosses Over Indicator 2) 
IF c1 THEN
 BUY 1 SHARES AT MARKET 
END IF
 
REM Venta
c2 = (Indicator 1 Crosses Under Indicator 2) 
IF c2 THEN
 SELLSHORT 1 SHARES AT MARKET 
END IF
 
REM Posicionamiento de un Stop dinámico 
IF Lowest [5] (Close) < (1.2 * Low) THEN
 IF Lowest [5] (Close) > = Close THEN
  Cut = Lowest [5] (Close)
 ELSE
  Cut = Lowest [20] (Close) 
 END IF
END IF
SET STOP Cut

Atención al uso correcto de órdenes STOP :

�� AT (precio) STOP define un nivel que lanza la orden en cuanto el precio lo toca.
�� SET STOP (precio) define un stop de protección, que cierra una posición existente.

SEGUIMIENTO DE POSICIONES

1. Instrucciones de veri�cación del estado de posiciones

ProBacktest permite subordinar la apertura de una posición a la existencia previa de otra posición en el mismo 
valor, tanto a largo como a descubierto. El control de las posiciones existentes se realiza directamente en el 
sistema, y el programa lo interpreta como una condición necesaria para que se abra una posición.

Los comandos que se utilizan para la gestión de estos casos son:

�� ONMARKET: comprueba si hay posiciones abiertas.
�� LONGONMARKET: comprueba si hay posiciones abiertas a largo.
�� SHORTONMARKET: comprueba si hay posiciones cortas (a descubierto) abiertas.

Se utilizan sin necesidad de paréntesis ni corchetes, y suelen incluirse habitualmente en las secciones condi-
cionales IF.

Las instrucciones de veri�cación del estado de posiciones son especialmente útiles en caso de acumulación 
(piramidación) de posiciones. Recordemos nuevamente que la acumulación de posiciones inversas sobre un 
mismo valor no es posible.

Veamos un ejemplo de uso de estos comandos:

Sistema en MACD (empleo de LONGONMARKET y de SHORTONMARKET):

El sistema que se muestra a continuación se basa en las inversiones del MACD respecto al nivel 0, y abrirá un 
cierto número de posiciones que se irán cerrando progresivamente. Esta evolución gradual presenta la ventaja 
de liquidar las ganancias y de limitar los riesgos.

REM De�nición del MACD
Indicator 1 = MACD [12,26,9] (Close)

REM Observación de los cambios de signo del Histograma delMACD 
c1 = (Indicator 1 Crosses Over 0.0)

REM Compra: Si no hay una posición larga abierta y si MACD >0 , compramos 3 títulos 
IF NOT LONGONMARKET AND c1 THEN
 BUY 3 SHARES AT MARKET ThisBarOnClose Long = 0
 Entry = Close 
END IF

REM Cierre posición larga: los 3 valores se venden sucesivamente a precios limite de 7%, 15% y 25%
REM de bene�cio si la evolución del precio lo permite
REM Las posiciones abiertas restantes se cierran cuando el MACD cruza el nivel 0 

IF LONGONMARKET AND Long = 0 AND Close > (Entry * 1.07) THEN
 SELL 1 SHARES AT MARKET ThisBarOnClose 
 Long = 1
ELSIF LONGONMARKET AND Long = 1 AND Close > (Entry * 1.15) THEN
 SELL 1 SHARES AT MARKET ThisBarOnClose 
 Long = 2
ELSIF LONGONMARKET AND Long = 2 AND Close > (Entry * 1.25) THEN
 SELL 1 SHARES AT MARKETThisBarOnClose 
 Long = 3
END IF

REM: Si no hay una posición corta (a descubierto) abierta y si MACD < 0, REM: vendemos a descubierto 3 
valores

IF NOT c1 AND NOT SHORTON MARKET THEN
 SELLSHORT 3 SHARES AT MARKET ThisBarOnClose
 Short = 0 
 Entry = Close
END IF

REM Cierre posición a descubierto: se compraran sucesivamente los 3 títulos a precios limite de
REM 7%, 15% y 25% de bene�cio si la evolución del precio lo permite
REM Las posiciones abiertas restantes se cierran cuando el MACD cruza el nivel 0

IF SHORTONMARKET AND Short = 0 AND Close < (Entry * 0.93) THEN
 EXITSHORT 1 SHARES AT MARKET ThisBarOnClose
 Short =1
ELSIF SHORTONMARKET AND Short = 1 AND Close < (Entry * 0.85) THEN 
 EXITSHORT 1 SHARES AT MARKET ThisBarOnClose
 Short =2
ELSIF SHORTONMARKET AND Short = 2 AND Close < (Entry * 0.75) THEN 
 EXITSHORT 1 SHARES AT MARKETThisBarOnClose
 Short =3
END IF

2. Contadores deposiciones

Para crear sistemas que tomen en cuenta la cantidad de órdenes pasadas, ProBacktest incluye comandos que 
cuentan las posiciones:

�� COUNTOFPOSITION: cantidad de posiciones abiertas desde el inicio del backtest
�� COUNTOFLONGSHARES: num. posiciones abiertas al alza (largas) desde el inicio del backtest
�� COUNTOFSHORTSHARES: num. posiciones abiertas a la baja (cortas) desde el inicio del backtest

Análogamente a las instrucciones de veri�cación del estado de posiciones, estos comandos se utilizan gene-
ralmente dentro de secciones condicionales.

Veamos a continuación un ejemplo que combina COUNTOFLONGSHARES y COUNTOFSHORTSHARES: 
Inverse Fisher Transform aplicado al RSI.

Este backtest se basa en el indicador ‘Inverse Fisher Transform RSI’ para dar órdenes de compra o de venta.

El sistema genera una orden de compra cuando el Inverse Fisher Transform RSI cruza al alza el nivel 50 y 
vende cuando el indicador cruza a la baja el nivel 80.

Inversamente, genera una orden de venta a descubierto si el Inverse Fisher Transform RSI cruza a la baja el 
nivel 50 y la cierra (compra) cuando dicho indicador cruza al alza el nivel 20.

El backtest debe utilizarse con velas de 1h en el caso de futuros. En cambio, es aconsejable utilizarlo con velas 
diarias en el caso de acciones.

REM Inverse Fisher Transform aplicado al RSI
REM Parámetros: n = Cantidad de velas a considerar en el calculo del RSI con un paso de 1
n = 10
Ind = RSI [n] (Close) x = 0.1 * (Ind -50)
y = (EXP (2 * x) - 1) / (EXP (2 * x) + 1) z = 50 * (y + 1) miInverseFisherTransformsRSI = z [7]
IF (miInverseFisherTransformsRSI Crosses Over 50) THEN 
 BUY 1 SHARES AT MARKET
END IF

IF (miInverseFisherTransformsRSI Crosses Under 80) THEN 
 SELL COUNTOFLONGSHARES SHARES ATMARKET
END IF

IF (miInverseFisherTransformsRSI Crosses Under 50) THEN
 SELLSHORT 1 SHARES AT MARKET 
END IF

IF (miInverseFisherTransformsRSI Crosses Over 20) THEN 
 EXITSHORT COUNTOFSHORTSHARES SHARES AT MARKET
END IF

3. ENTRYINDEX

El comando ENTRYINDEX[.] da acceso a la vela de la enésima transacción previa. Presenta características 
similares a las de las funciones BarIndex e IntradayBarIndex (presentadas en el manual ProBuilder):

�� La búsqueda se realiza analizando el gráfico de izquierda a derecha.
�� La primera vela mostrada a la izquierda se entiende como vela 0 (p.ej., ENTRYINDEX = 3 sería la cuarta 
vela tras el inicio del histórico).

La sintaxis del comando es similar a la de una constante:

ENTRYINDEX [enésima orden previa]

Nota: ENTRYINDEX puede emplearse sin asociación a un número de vela de�nido entre corchetes. En tal caso, 
el programa tomará en consideración la vela de la última orden generada.

Ejemplo: Find InsideBar

Este ejemplo es un sistema basado en un popular patrón de precios conocido como ‘Inside Bar’. El sistema:

�� Evalúa si el rango de la tercera vela (contando la vela actual –considerada como vela 0) es superior al rango 
de la vela siguiente, y si ésta última es blanca (o verde). En tal caso, se abre una posición a largo (se genera 
una orden de compra)

�� Evalúa si el rango de la tercera vela (contando la vela actual –considerada como vela 0) es inferior al rango 
de la vela siguiente, y si ésta última es negra (o roja). En tal caso, se abre una posición a corto (se genera 
una orden de venta a descubierto).

La posición se cierra sistemáticamente 3 velas después de haber sido abierta.

Condicion 1 = (High [2] > = High [1] AND Low [2] < = Low [1]) 
Condicion 2 = (High [2] < = High [1] AND Low [2] < = Low [1]) 
Condicion 3 = (Close [1] > Open [1])
Condicion 4 = (Close [1] < Open [1])

IF (Condicion 1 AND Condicion 3) THEN
 BUY 10 %CAPITAL AT MARKET NextBarOpen 
END IF

IF LONG ON MARKET AND (BarIndex - ENTRYINDEX) = 3 THEN 
 SELL 10 %CAPITAL AT MARKET ThisBarOnClose
ENDIF

IF (Condicion 2 AND Condicion 4) THEN
 SELL SHORT 10 %CAPITAL AT MARKET NextBarOpen 
END IF



PROBACKTEST

Página 25

IF SHORT ON MARKET AND (BarIndex - ENTRYINDEX) = 3 THEN
 EXIT SHORT AT MARKET ThisBarOnClose 
END IF

4. ENTRYQUOTE

El comando ENTRYQUOTE [.] llama al precio al que una orden se ha ejecutado. Es particularmente relevante 
cuando la duración entre cada operación es corta; es decir, en sistemas intraday. Su sintaxises:

ENTRYQUOTE [enésima orden previa]

Como para todas las constantes, podemos indicar entre corchetes la orden a la que nos referimos. Si no se indica 
nada tras el comando ENTRYQUOTE, el programa tomará en consideración la vela de la última orden generada.

Ejemplo: Creación de un target (objetivo de bene�cios o take-pro�t) De�nimos dos condiciones:

�� Ausencia de posiciones abiertas.
�� RSI débil (<30).

La compra tendrá lugar cuando el precio sea superior a la media móvil de período 10.

La posición se cerrará cuando el precio en tiempo real supere (hemos usado una orden límite) al precio de 
compra en un 15%.

IF NOT ON MARKET AND RSI < 30 THEN
 IF Close > AVERAGE [10] (Close) THEN 
  BUY 100 %CAPITAL AT MARKET
 END IF 
END IF

SELL 100 %CAPITAL AT ENTRY QUOTE * 1.15 LIMIT

5. PreviousTrade

Algunos traders consideran los resultados anteriores para construir sus sistemas. El  comando

PreviousTrade (.) sirve precisamente para llamar a estos resultados.

Representa los resultados en % de la enésima operación realizada. Su sintaxis de uso es análoga a la de una 
constante diaria deprecio:

PreviousTrade (enésima operación previa)

PreviousTrade (1) = Nivel de la última operación realizada. ¡Cuidado con los paréntesis, son obligatorios!

Ejemplo:

Veamos un ejemplo basado en el cruce de las líneas del estocástico y del indicador RSI. Comenzamos por 
enviar una orden de compra a precio de mercado cuando las medias móviles exponenciales se cruzan; a 
continuación, se abrirá una posición:

�� de compra (acumulando posiciones) si:
�� la primera orden enviada está en resultado positivo.
�� RSI es <30.

�� de venta si:
�� RSI es> 70.
�� la línea K cruza a la baja a la línea D.

ONCE StochPeriod = 14 
ONCE KPeriod = 3 ONCE 
DPeriod = 3
LineK = Stochastic [StochPeriod, KPeriod] (Close) 
LineD = Average [DPeriod] (LineK)

// Envío de la primera orden a precio demercado
IF ExponentialAverage [12] (Close) Crosses Over ExponentialAverage [20] (Close) THEN
 BUY AT MARKET 
END IF

//Envío de la segunda orden
IF LineK Crosses Over LineD THEN 
 IF RSI < 30 THEN
  REM Compra si la orden anterior se encuentra en bene�cios latentes 
  IF PreviousTrade (1) > 0 THEN
   BUY 100 %LIQUIDITY AT MARKET 
  END IF
 END IF 
END IF

IF RSI > 70 AND LineK Crosses Under LineD THEN 
 SELL 100 %CAPITAL AT MARKET
END IF

CAPITULO III: APLICACIONES PRÁCTICAS

SISTEMAS SOBRE INDICADORES

1. Sistema basado en los Heikin Ashi

Este sistema genera una señal de compra cuando en estilo Heiken-Ashi aparece una vela roja seguida de una 
vela verde.

Inversamente, se genera una señal de venta a descubierto si aparece una vela verde seguida de una roja.

El interés de este backtest se relaciona con el método de construcción de la vista Heikin Ashi, en la que no 
es posible aplicar sistemas. Por ello, este ProBacktest deberá aplicarse imperativamente en un grá�co con el 
precio en estilo velas (candle sticks).

ONCE PreviousStatus = 0 

IF BarIndex = 0 THEN
 XClose = TotalPrice
 XOpen = (Open + Close) /2
ELSE
 XClose = TotalPrice
 XOpen = (XOpen [1] + Xclose [1]) / 2 
END IF

IF XClose > = XOpen THEN
 IF PreviousStatus = -1 THEN
  BUY 1 SHARES AT MARKET
 ELSE
  PreviousStatus = 1
  IF PreviousStatus = 1 THEN
   SELLSHORT 1 SHARES AT MARKET
   PreviousStatus = -1 
  END IF
 END IF
END IF

2. Sistema basado en el ZigZag

Este backtest utiliza el indicador Zig Zag para identi�car cuales habrían sido las mejores oportunidades e com-
pra y venta. Los excelentes resultados (en mercados de acciones y futuros) se deben al carácter no predictivo 
del Zig Zag, que se recalcula a posteriori y no proporciona sistemáticamente señales válidas en tiempo real. 
A pesar de ello, este sistema interesa en tanto que permite comparar sus resultados con los de otros sistemas.

// En variable optimizada: periodos = 1 (de 5 à 10) 
c11 = (myZigZag > myZigZag [1])
c12 = (myZigZag < myZigZag [1])

IF c11 AND (SHORTONMARKET OR NOT LONGONMARKET) THEN 
 EXITSHORT COUNTOFSHORTSHARES SHARES AT MARKET 
 BUY 50 %CAPITAL AT MARKET
END IF

IF c12 AND (LONGONMARKET OR NOT SHORTONMARKET) THEN 
 SELL COUNTOFLONGSHARES SHARES AT MARKET 
 SELLSHORT 50 %CAPITAL ATMARKET
END IF

3. Sistema Breakout Range con Stop Dinámico

Se trata de un sistema de tipo BreakOut: las señales se generan a partir de rupturas de máximos o de mínimos 
calculadas a lo largo de un período determinado. Este sistema abre únicamente posiciones largas que protege 
a continuación con un stop dinámico. El período sobre el cual se calcula el máximo debe determinarse como 
una variable optimizada.

REM periodo = variable optimizada (de 2 a 20 en pasos de 1) 
ONCE MMentry = 5
ONCE Period = 14 

REM Entrada a largo:
Condition = High > Highest [Period] (High) [1]

IF Condition AND Summation [Period] (Condition) = 1 THEN 
 BUY 1 SHARES AT MARKET
END IF

c2 = Lowest [10] (Low [1])
StopLoss = Highest [MMentry] (High) [BarIndex - ENTRYINDEX + 1] / Average [20] (High/Low)
SET STOP MAX (StopLoss, (c2 -0.01))

4. Sistema basado en el Estocástico Alisado

Este sistema reposa sobre el indicador ‘Estocástico alisado’ aplicado al precio mediano y su media móvil.

La señal de compra aparece cuando el indicador es superior a su media móvil exponencial. Se de�ne un 
objetivo de tipo Límite, superior en un 10% al precio de compra. La señal de venta  se da cuando el indicador 
desciende nuevamente por debajo de su media móvil exponencial.

REM Compra
Indicator 1 = SmoothedStochastic [9,9] (MedianPrice) 
Indicator 2 = ExponentialAverage [9] (Indicator1)

// Inicio variable 
StopLimit = 10
c1 = (Indicator 1 > = Indicator 2) 
IF c1 THEN
 BUY 1 SHARES AT MARKET RealTime 
END IF

IF LONGONMARKET THEN
 SELL AT ENTRYQUOTE * (1 + StopLimit/100) Limit 
END IF

IF SHORTONMARKET THEN
 EXITSHORT AT ENTRYQUOTE / (1 + StopLimit/100) Limit 
END IF

REMVenta
IF NOT c1 THEN
 SELL AT MARKET RealTime 
END IF

REM Venta a descubierto 
IF NOT c1 THEN
 SELLSHORT 1 SHARES AT MARKET RealTime 
END IF

REM Recompra 
IF c1 THEN
 EXITSHORT AT MARKET RealTime
END IF

5. Swing Trading, ADX y Medias móviles

Este backtest se basa en el indicador ADX y en su posicionamiento respecto al valor 30, con el objetivo de 
reducir las señales falsas y minimizar los riesgos.

Se trata de un sistema que presenta un gran número de condiciones, lo que restringe el número de señales.

MyADX12 = ADX [12]
ADXperiods = 5
MyMM20 = Average [20] (Close) IsLow30 = 0
FOR Count = 0 TO ADXperiods DO
 IF MyADX12 [Count] < 30.0 THEN
  IsLow30 = 1 
  BREAK
 END IF
NEXT

// Compra

// ADX 12 esta por encima de 30 desde al menos entre 5 y 10 sesiones 
Condition 1 = NOT IsLow30

// Si los precios vuelven a apoyarse en la MME 20 días durante 1 a 4 sesiones consecutivas
Condition 2 = High > MyMM20 AND Low < MyMM20 AND High [1] < MyMM20 [1] AND Low [1] < MyMM20 [1]

// Si el máximo del día sobrepasa el máximo de la víspera 
Condition 3 = Dhigh (0) > Dhigh (1)
IF Condition 1 AND Condition 2 AND Condition 3 THEN 
 BUY 1 SHARES AT MARKET ThisBarOnClose
END IF

// Venta acorto

// ADX 12 esta por encima de 30 desde al menos entre 5 y 10 sesiones 
Condition 4 = NOT IsLow30

// Si los precios vuelven a apoyarse en la MME 20 días durante 1 a 4 sesiones consecutivas
Condition 5 = High > MyMM20 AND Low < MyMM20 AND High [1] > MyMM20 [1] AND Low [1] > 
MyMM20 [1]

// Si el mínimo del día sobrepasa el mínimo de lavíspera
Condition 6 = Dlow (0) < Dlow (1)

IF Condition 4 AND Condition 5 AND Condition 6 THEN 
 SELLSHORT 1 SHARES AT MARKET ThisBarOnClose
END IF

SISTEMAS DE MONEY MANAGEMENT (GESTIÓN DE CAPITAL)

Los resultados de un backtest pueden mejorarse sustancialmente adoptando reglas so�sticadas de gestión de 
capital (money management).

Estas estrategias de money management suelen formalizarse en las martingalas, destinadas a optimizar la 
esperanza matemática de un sistema (ganancia o pérdida media por operación). Esto implica poder estimar 
previamente la probabilidad de que una operación resulte en ganancia, y el importe del bene�cio.

Puede ser interesante codi�car directamente órdenes stop (de protección), take pro�t (objetivo) e inactividad, 
así como subsistemas que permitan la gestión dinámica del tamaño de posiciones.

1. Stop de protección (stoploss)

El código presentado a continuación integra una orden stop loss de tipo stop limit en su sistema. Recuerde 
de�nir las condiciones de su stop; en este caso se les denomina StopLossLong y StopLossShort.

ONCE Level = 0.05
REM Selección del nivel de perdida máxima determinante del cierre de posición (0.05 equivale a 5%)

REM Si la posición es a largo, el cierre se da cuando el precio varia de -nivel% (bajada de-nivel %)
IF LONGONMARKET AND (Close - ENTRYQUOTE) / (ENTRYQUOTE) < Level THEN
 SELL AT MARKET StopLossLong 
END IF

REM Si la posición es a corto, el cierre se da cuando el precio varía de nivel% (subida de nivel %)
IF SHORTONMARKET AND (Close - ENTRYQUOTE) / (ENTRYQUOTE) > Level THEN
 EXITSHORT AT MARKET StopLossShort 
END IF

2. Objetivo de bene�cio (takepro�t)

Este código integra un objetivo (take pro�t) de tipo StopLimit en su sistema. Recuerde de�nir las condiciones de 
su stop; en este caso se les denomina TakePro�tLong y TakePro�tShort.

ONCE Level = 0.05
REM  Selección del nivel de ganancia determinante del cierre de posición (0.05 equivale a 5%)

REM Si la posición es a largo, el cierre se da cuando el precio varía de nivel% (subida de nivel %)
IF LONGONMARKET AND (Close - ENTRYQUOTE) / (ENTRYQUOTE) > Level THEN
 SELL AT MARKET TakePro�tLong 
END IF

REM Si la posición es a corto, el cierre se da cuando el precio varia de -nivel% (bajada de-nivel %)

IF SHORTONMARKET AND (Close - ENTRYQUOTE) / (ENTRYQUOTE) < Level THEN
 EXITSHORT AT MARKET TakePro�tShort 
END IF

3. Stop de inactividad

Este código permite integrar un stop de inactividad en su sistema. Recuerde de�nir las condiciones de su stop; 
en este caso se les denomina InactivityStopLong e InactivityStopShort. En el ejemplo descrito, el stop se activa 
tras 10 velas.

ONCE Count =10
REM Seleccionar el numero de velas a contar desde la apertura de posición, tras el cual dicha 
REM posición se cerrará sistemáticamente
IF ONMARKET AND (BarIndex - ENTRYINDEX + 1) > Count THEN 
 IF LONGONMARKET THEN
  SELL AT MARKET InactivityStopLong 
 END IF

 IF SHORTONMARKET THEN
  EXITSHORT AT MARKET InactivityStopShort 
 END IF
END IF

4. Piramidación de una posición

Para acumular posiciones es necesario activar la casilla ‘Acumular posiciones’ en la ventana ‘Gestión capital’ 
del backtest. La piramidación consiste en el envío de varias órdenes sucesivas en la misma dirección, con el 
objetivo de aumentar el tamaño de la posición. El piramidaje se concreta en el momento en que varias órdenes 
se validan simultáneamente, tal y como se muestra en este sencillo backtest:

REM Compra 1 cuando el RSI es inferior a 30 
IF RSI [14] (Close) < 30 THEN
 BUY 1 SHARES AT MARKET 
END IF

REM Si se ha abierto una posición larga (compra) y el precio al cierre de la vela anterior esta
REM por debajo del precio de apertura de la vela actual, se compra cada vez 1 valor suplementario
IF LONGONMARKET AND Open > Close [1] THEN 
 BUY 1 SHARES AT MARKET
END IF

REM Cuando el precio cruza a la baja una media móvil simple, se cierra la totalidad  de la posición
IF Close Crosses Under Average [14] (Close) THEN 
 SELL 100 %CAPITAL AT MARKET
END IF

5. Gestión dinámica del tamaño de órdenes

Para modi�car dinámicamente el tamaño de las posiciones sin pasar obligatoriamente por el piramidaje, es 
posible utilizar una variable que indica la cantidad de partes del activo subyacente que se deseen negociar.

ONCE OrderSize = 1
REM Las órdenes son inicialmente sobre 1 parte del activo subyacente 
BUY OrderSize SHARES AT MARKET

REM la posición se cierra tras 2 velas 
IF BarIndex - ENTRYINDEX > = 2 THEN
 SELL AT MARKET 
END IF

REM Si le RSI es inferior a 30, se aumenta la posición en cada vela REM 
Este tamaño se limita a un máximo de 50
IF RSI [14] (Close) < 30 THEN
 OrderSize = MAX (OrderSize, (OrderSize + 1.50)) 
END IF

REM Si el RSI es superior a 70, se disminuye en cada vela el tamaño de la posición a abrir
REM Este tamaño no puede ser inferior a 0 
IF RSI [14] (Close) > 70 THEN
 OrderSize = MIN (OrderSize, (OrderSize -1.0))
END IF

 
6. Consideración de los resultados históricos

Mediante la variable PreviousTrade(i) se puede alterar el comportamiento del sistema presentado a continuación, 
independientemente de que resulte en pérdidas o en bene�cios. Retomando el backtest previo, es posible mejorar 
su comportamiento reduciendo el tamaño de las posiciones en caso de pérdida (dado que en tal caso no estaría 
siguiendo al mercado correctamente), o al contrario, aumentándolo cuando las ganancias se repiten.

ONCE OrderSize = 1
REM Las ordenes son inicialmente sobre 1 parte del activo subyacente 
BUY OrderSize SHARES AT MARKET
REM la posición se cierra tras 2 velas IF BarIndex - ENTRYINDEX >= 2 THEN
 SELL AT MARKET 
END IF

REM Si le RSI es inferior a 30, se aumenta la posición en cada vela REM Este tamaño se limita a un máximo 
de 50
IF RSI [14] (Close) < 30 THEN
 OrderSize = MAX (OrderSize, (OrderSize + 1.50)) 
END IF

REM Si el RSI es superior a 70, se disminuye en cada vela el tamaño de la posición a abrir
REM Este tamaño no puede ser inferior a 0 
IF RSI [14] (Close) > 70 THEN
 OrderSize = MIN (OrderSize, (OrderSize - 1.0)) 
END IF

REM Modi�camos el comportamiento del backtest en función de los resultados pasados REM analizando suce-
sivamente las 3 ultimas transacciones.
REM Si una operación da perdidas, el tamaño de la posición cae en 1. Si da  ganancias, aumenta de1
FOR i = 1 TO 3 DO
 IF PreviousTrade (i) >= 0 THEN
  OrderSize = MAX (OrderSize, (OrderSize + 1.50)) 
 ELSIF PreviousTrade (i) < 0 THEN
  OrderSize = MIN (OrderSize, (OrderSize - 1.0)) 
 END IF
NEXT

Con estas herramientas se pueden integrar varias martingalas en los sistemas de ProBacktest. He aquí algunos 
ejemplos de técnicas populares de gestión del capital susceptibles de integrar otras estrategias..

7. La martingala clásica

La martingala clásica consiste en doblar la posición cuando se afronta una pérdida, con la idea de reembol-
sar en la siguiente transacción y ganar tras recuperar la posición de salida. El  mayor inconveniente de esta 
gestión del capital es que si se producen varias pérdidas consecutivas, la posibilidad de duplicar la posición 
se hace cada vez menos factible. Así, partiendo de un capital de p.ej. 1000€, cinco pérdidas sucesivas re-
querirían 1000 x 25 euros = 32000€ para continuar con el sistema.

Los sistemas derivados de esta martingala pueden ser más adaptables a la operativa con acciones que con 
futuros o Forex, ya que la inversión inicial y el apalancamiento suele ser más importante en estos últimos mer-
cados. Para aplicarla, se integrará este código en las condiciones de entrada y de salida:

//*********** Codigo a insertar al principio del sistema **********//

ONCE OrderSize = 1
REM Inicio con una posición de 1

//*********************//

//*********** Código a insertar tras las instrucciones que cierran una posición **********//

IF PreviousTrade (1) < 0 THEN
 OrderSize = OrderSize *2

REM Si la última transacción resulta en perdidas, se dobla el tamaño de la posición

ELSIF PreviousTrade(1) > 0 THEN 
 OrderSize = 1
REM Si la última transacción resulta en ganancias, se retorna a una posición de tamaño 1

END IF

//*********************//

REM En el backtest, el tamaño de la posición queda de�nido por la variable OrderSize

8. La gran martingala

La gran martingala es similar a la clásica, con la única diferencia que además de doblar la posición en cada 
pérdida, se añade una unidad suplementaria.

Esta martingala es más arriesgada que la clásica en caso de pérdidas sucesivas, pero permite aumentar sig-
ni�cativamente las ganancias en caso contrario. Para aplicarla, se integrará este código en las condiciones 
de entrada y de salida:

//*********** Codigo a insertar al principio del sistema **********//

ONCE OrderSize = 1
REM Inicio con una posición de 1

//*********************//

//*********** Código a insertar tras las instrucciones que cierran una posición **********//

IF PreviousTrade (1) < 0 THEN
 OrderSize = OrderSize * 2 + 1
// Si la última transacción resulta en perdidas, se dobla el tamaño de la posición y se añade una unidad

ELSIF PreviousTrade (1) > = 0 THEN 
 OrderSize = 1
// Si la última transacción resulta en ganancias, se retorna a una posición de tamaño 1

END IF

//*********************//

REM En el backtest, el tamaño de la posición queda de�nido por la variable OrderSize

9. La Piquemouche

La Piquemouche es otra variante de la martingala clásica. En caso de pérdida, se aumenta el tamaño de la 
posición en 1 si se han dado menos de 3 pérdidas consecutivas. Cuando se acumulan más de 3 pérdidas se-
guidas, se duplica el tamaño de la posición. La primera ganancia reinicia el tamaño de la posición, �jándolo 
nuevamente a 1.

Este sistema de gestión de posiciones es menos arriesgado que los dos anteriores, ya que retrasa el aumento 
del tamaño de la posición hasta pasadas las 3 pérdidas sucesivas. Para aplicarla, se integrará este código en 
las condiciones de entrada y de salida:

//*********** Código a insertar al principio del sistema **********//

ONCE OrderSize = 1
REM Inicio con una posición de 1

ONCE BadTrades = 0
// Inicialización del contador de transacciones sucesivas con resultado de pérdidas

//*********************//

//*********** Código a insertar tras las instrucciones que cierran una posición **********//

IF PreviousTrade (1) < 0 THEN
 BadTrades = BadTrades + 1 
  IF BadTrades < 3 THEN
//Si la ultima operación resulta en pérdida y no se superan 3 seguidas, se aumenta en 1 unidad el tamaño 
de la próxima posición

   OrderSize = OrderSize + 1
  ELSIF PreviousTrade (1) < 0 AND BadTrades MOD 3 = 0 THEN
// Si la ultima operación resulta en perdida y se acumulan mas de 3 seguidas, se duplica el tamaño de la 
próxima posición

   OrderSize = OrderSize * 2 
  END IF

ELSIF PreviousTrade (1) > = 0 THEN
// Si la ultima operación resulta en ganancia, la próxima posición volverá a tamaño 1

 OrderSize =1
 BadTrades =0
END IF

//*********************//
REM En el backtest, el tamaño de la posición queda de�nido por la variable OrderSize

10. La Whittacker

La Whittacker consiste en con�gurar el tamaño de la próxima posición igual a la suma del tamaño de las dos 
posiciones previas en caso de pérdida. Si hay una ganancia, se reinicia a 1  el tamaño de la posición. Para 
aplicarla, se integrará este código en las condiciones de entrada y de salida:

 //*********** Código a insertar al principio del sistema **********//

ONCE OrderSize = 1
// Inicio con una posición de 1

//*********************//

//*********** Código a insertar tras las instrucciones que cierran una posición **********//

IF PreviousTrade (1) < 0 THEN
 OrderSize = OrderSize + OrderSize [1] 

ELSIF PreviousTrade (1) > = 0 THEN
 OrderSize = 1
// Si la última operación resulta en ganancia, la próxima posición volverá a tamaño 1

END IF

//*********************//
REM En el backtest, el tamaño de la posición queda de�nido por la variable OrderSize

11. La pirámide de Alembert

Concebida por d’Alembert (matemático francés del siglo XVIII), esta martingala aumenta la posición de 1 n 
caso de pérdida; en caso de ganancia, la disminuye de 1. Esta técnica sólo es pertinente cuando se considere 
que las ganancias sucesivas disminuyen la probabilidad de que la siguiente operación resulte en ganancia (y 
recíprocamente, que una pérdida aumente la posibilidad de que la siguiente operación resulte en ganancia). 
Para aplicarla, se integrará este código en las condiciones de entrada y de salida:

//*********** Código a insertar al principio del sistema **********//

ONCE OrderSize = 1
// Inicio con una posición de 1

//*********************//

//*********** Código a insertar tras las instrucciones que cierran una posición **********//

IF PreviousTrade (1) < 0 THEN

 OrderSize = OrderSize + 1 

ELSIF PreviousTrade (1) > = 0THEN
 OrderSize = MAX (OrderSize, (OrderSize -1.1))
END IF

12. La contra de Alembert

Esta es la estrategia recíproca de la pirámide epónima, ya que se disminuye el tamaño de la posición de 1 en 
caso de pérdida y se aumenta de 1 en caso de ganancia.

La técnica implica la consideración de que los resultados históricos son representativos de los resultados 
futuros.

Para aplicarla, se integrará este código en las condiciones de entrada y de salida:

//*********** Código a insertar al principio del sistema **********//

ONCE OrderSize = 1
// Inicio con una posición de 1

//*********************//

//*********** Código a insertar tras las instrucciones que cierran una posición **********//

IF PreviousTrade (1) < 0 THEN
 OrderSize = MAX (OrderSize, (OrderSize - 1.1)) 

ELSIF PreviousTrade (1) > = 0 THEN
 OrderSize = OrderSize + 1 
END IF

//*********************//

ANEXO: CONFIGURAR LA GESTIÓN DEL CAPITAL

La ‘gestión de capital’ se encuentra disponible en la sección Money Management de la ventana de programa-
ción.

Es un elemento clave que puede in�uir decisivamente sobre el resultado �nal del sistema; así, una disminución 
de las comisiones de operativa, o una política más relajada de gestión del riesgo, podría resultar en una sen-
sible variación del resultado neto.

Se compone de 5 secciones con�gurables:

�� Capital.
�� Comisiones de operativa.
�� Gestión de riesgos (risk management).
�� Gestión de posiciones.
�� Gestión del redondeo de la cantidad de valores a comprar/vender.

Existen reglas especí�cas aplicables a la sección de comisiones de operativa al trabajar sobre distintos tipos 
de valores (acciones, futuros o divisas). Por ello, trataremos estas comisiones al �nal del presente anexo.

CAPITAL

En esta sección, basta con indicar la suma que desea invertir mediante el sistema.

Nota: salvo mención explícita (ver la con�guración del límite máximo del importe invertido), el ProBacktest 
dejará de abrir posiciones cuando el capital inicial se haya agotado.

Si su sistema no abre posiciones, pruebe a aumentar el importe del Capital inicial.

GESTIÓN DE RIESGOS (RISK MANAGEMENT)

La sección permite de�nir el límite máximo de inversión en el mercado a través de  3 secciones:

1. Límite máximo del total invertido

Este campo limita las pérdidas y gestiona el apalancamiento. Empiece seleccionando en el menú desplegable 
el modo de inversión del capital. Preste atención a la elección de la escala: valor absoluto (en efectivo), % 
capital o % liquidez.

2. Montante máximo por transacción y apalancamientos

La inversión máxima por operación le permite limitar la cantidad invertida en el envío de una orden. Funciona 
de modo similar a la descrita previamente.

Gestione el apalancamiento mediante la combinación de ambas funciones. Veamos un  ejemplo:

�� Límite máximo de inversión: 500% Capital.
�� Inversión máxima por operación: 500% Capital.

A través de esta con�guración, cada orden enviada tendrá un apalancamiento de 5. Inversamente, si de�ne 
un % de 100, limitará las pérdidas al valor total de su cartera.

3. Montante mínimo por transacción

La principal utilidad del montante mínimo por transacción es de evitar que las posiciones abiertas sean dema-
siado pequeñas, previniendo comisiones de operativa desmesuradas respecto al importe invertido.

Ejemplo: la compra de una acción XXX que cotiza a 5€ con comisiones de 5€ por orden, implicaría que las 
comisiones representan el 100% del importe invertido. Ello implica comenzar la inversión con una brecha 
relativa de -100%, que se transformaría en -200% si se  consideran además las comisiones de una orden de 
venta al mismo precio.

GESTIÓN DE POSICIONES

En esta sección tiene un máximo de 3 casillas a seleccionar.

Reinvertir bene�cios con�gura la gestión de las ganancias. Tras el cierre de una posición,  por defecto el 
sistema no aumenta el capital inicial con los bene�cios que se hayan podido obtener. En cambio, si selecciona 
esta opción, los bene�cios aumentarán el capital disponible para invertir.

Acumular posiciones de�ne si las posiciones abiertas se deben acumular entre sí en caso de que las condicio-
nes de entrada se cumplan repetidamente, pero sin permitir el cierre de una posición ya abierta. Por defecto 
se abre una sola posición cada vez, a menos que en el código haya una indicación explícita para actuar de 
otro modo. En tal caso, la activación de esta opción no in�uiría sobre los resultados del sistema.

Si ha creado stops prede�nidos en la interfaz, puede asociarlos a las posiciones según estos parámetros:

�� Un stop para todas las posiciones (aplicable en el caso de acumulación de posiciones).
�� Un stop por posición.

Esta sección es particularmente interesante si se aplica a un stop dinámico. Tratándose de un stop que sigue 
la evolución del precio y que se de�ne habitualmente como la distancia en %  del último precio, en el caso de 
acumulación de posiciones el sistema podrá elegir entre aplicar un único stop dinámico (aplicado al precio 
medio ponderado de las posiciones abiertas) o aplicar un stop dinámico por cada posición abierta.

REDONDEAR LA CANTIDAD DE VALORES A COMPRAR /VENDER

Esta sección es relativamente simple a rellenar. Permite de�nir cómo se hará el redondeo de la cantidad de 
valores a comprar o a vender, y se realiza mediante la selección de una de estas alternativas:

COMISIONES DE OPERATIVA EN ACCIONES

Para aplicar un ProBacktest a instrumentos de tipo acción o similar (warrants, ...), deberá activarse la pestaña 
‘Por orden’ en la sección ‘Comisión’. Tal y como se ilustra en la �gura a continuación, se pueden con�gurar 
las comisiones de operativa en precio real o en %.

Las comisiones de operativa por orden sólo se aplican en la apertura de la posición. Una operativa en ambos 
sentidos (compra/venta a largo o venta a descubierto/recompra a corto)  se corresponde con dos órdenes -y 
por lo tanto, con el doble de comisiones por orden.

La casilla ‘Comisión por orden’ de�ne la cantidad mínima de las comisiones en %.

Ejemplo: si las comisiones mínimas por orden son de 15€ y el % de la transacción sólo asciende a 10€, las 
comisiones de operativa totales serían de 15€.

1. Comisiones de operativa en futuros

Para con�gurarlas se debe pulsar en la pestaña ‘Por contrato (futuros)’ de la sección  ‘Comisión’. A continua-
ción, podrá de�nir:

�� La comisión por contrato.
�� El depósito por contrato.
�� El valor de un punto.

a. Comisión por lote
En los mercados de futuros, la comisión por lote o contrato puede considerarse como equivalente a las comisio-
nes de operativa en acciones. Depende de cada intermediario �nanciero, quien de�ne e informa del montante 
aplicable.

b. Depósito por lote
El depósito por lote constituye una garantía para el broker. Entre otras cosas,  permite  gestionar el apalanca-
miento (elemento importante en ciertos futuros).

Ejemplo:

Dispone de 500€ para invertir, y su broker le autoriza a invertir hasta un total de 1000€. Los 500€ iniciales 
servirán de depósito para su broker. Con 1000€ dispone de un apalancamiento  de 2. Si se intenta replicar 
al mercado lo más �elmente posible, deberá considerarse que el broker no esperará a que se produzcan 
pérdidas de 1000€ para liquidar sus posiciones. En este ejemplo, podríamos estimar que dicho cierre de 
posiciones tendrá lugar en el momento en que el depósito sea insu�ciente para cubrir las pérdidas (se cierran 
las posiciones a partir de 500€ de pérdidas).

El depósito por contrato varía según el valor negociado y según el broker. Para conocer el importe preciso, 
será necesario consultar con su propio intermediario �nanciero.

No obstante, si desea simular un cierto apalancamiento, le ofrecemos la fórmula que le permitirá calcular el 
depósito por contrato respecto a dicho apalancamiento:

Depósito por lote = Capital inicial / Apalancamiento

Supongamos que dispone de un capital inicial de 1000€ y que desea disponer de un apalancamiento de 5. 
Su depósito deberá ser de: 1000/5 = 200€.

c. Valor de un punto
En la plataforma, este valor indica el apalancamiento aplicable sobre las ganancias o pérdidas realizadas. 
Depende del futuro en el que se aplica el backtest, y se expresa en unidades monetarias por unidad de valor 
del futuro. Se calcula mediante la siguiente fórmula:

Valor de un punto = Valor monetario de un tick / Tamaño de un tick

Siendo el tick la variación mínima del futuro autorizada por el mercado donde cotiza (también llamado ticksize).

He aquí la tabla con el ticksize de los futuros más importantes:

Nombre del Futuro Valor 1tick Tamaño 1tick Valor 1punto

FCE CAC40 5€ 0,5 10€

DAX 12,5€ 0,5 25€

DJ Eurostoxx50 10€ 1 10€

BUND 10€ 0,01 1000€

EuroFX 12,5$ 0,0001 125000$

Mini S&P500 12,5$ 0,25 50$

Mini Nasdaq100 5$ 0,25 20$

Mini Dow 5$ 1 5$

2. Comisiones de operativa en Forex (divisas)

Empiece pulsando en la pestaña ‘Por contrato (futuros)’ de la sección ’Comisión’.

Deberá rellenar los tres campos ya analizados en el caso de futuros, siguiendo estas indicaciones:

a. Comisión por lote/Spread
La comisión por lote es el equivalente FOREX del spread multiplicado por el valor de un pip y dividido por dos 
(el spread es la comisión abonada por una operación de compraventa). La noción de comisión puede parecer 
extraña en el caso del FOREX; generalmente, se emplea el término SPREAD. Este corresponde a la diferencia 
entre los precios de venta (Bid) y de compra (Ask):

SPREAD = (Bid -Ask)
Comisión por lote = Spread * Valor de un Pip /2

Sin embargo, cada intermediario gestiona sus spreads de manera propia (distintos posicionamientos del Bid y 
del Ask). Además, los spreads de un mismo intermediario pueden variar en la variación del cruce, o cambiar 
en el tiempo. Por ello, normalmente no suele ser posible calcular el spread de su intermediario en la ope-
rativa de divisas, por lo que los resultados de los backtest realizados en este tipo de valores no deberán ser 
interpretados como una información totalmente realista. Deberá considerarse una pequeña brecha debida a 
las variaciones aplicadas al spread a lo largo del tiempo.

Por ello, nuestra sugestión es dejar este campo con un valor 0.

En cambio, si desea simular el spread aplicado por su intermediario, bastará con insertar la cantidad de pips 
* valor de un pip /2.

En el caso de un spread igual a 3 pips (caso del ’EURUSD), podría con�gurar una comisión por lote de 3 * 
10$ / 2 = 15$.

b. Depósito por lote
El depósito por lote constituye una garantía para el intermediario. Permite gestionar el apalancamiento, que 
se revela de especial importancia en ciertas paridades como las japonesas.

Ejemplo:

Dispone de 500€ para invertir, y su broker le autoriza a invertir hasta un total de 1000€. Los 500€ iniciales 
servirán de depósito para su broker. Con 1000€ dispone de un apalancamiento  de 2. Si se intenta replicar 
al mercado lo más �elmente posible, deberá considerarse que el broker no esperará a que se produzcan 
pérdidas de 1000€ para liquidar sus posiciones. En este ejemplo, podríamos estimar que dicho cierre de 
posiciones tendrá lugar en el momento en que el depósito sea insu�ciente para cubrir las pérdidas (se cierran 
las posiciones a partir de 500€ de pérdidas).

El depósito por contrato varía según el valor negociado y según el broker. Para conocer el importe preciso, 
será necesario consultar con su propio intermediario �nanciero.

No obstante, si desea simular un cierto apalancamiento, le ofrecemos la fórmula que le permitirá calcular el 
depósito por contrato respecto a dicho apalancamiento:

Valor de un punto = Valor monetario de un tick / Tamaño de un tick

Supongamos que desea operar en la paridad EUR/USD (lote de 100 000$) con un apalancamiento de 100. 
Su depósito por lote será: 100 000$ / 100 = 1 000€.

c. Valor de un punto
En la plataforma, este valor indica el apalancamiento aplicable a los bene�cios o a las pérdidas realizados. 
Depende del par de divisas sobre el que se aplique el backtest, y se expresa en la unidad monetaria del valor 
del pip.

El valor por punto se calcula mediante la siguiente fórmula:

1 punto = Valor de un Pip / Paso de cotización de la paridad

Ejemplo:

Sea el EURUSD cuyo paso de cotización es de 0,0001. Sabemos que 1 pip =10$.

Para el EURUSD, un lote de tamaño estándar es 100 000$: 1 punto = 10 / 0,0001 = 100000$.

